4　要素剛性マトリックスの重ね合わせ

全体座標系で表された各要素の剛性マトリックスを重ね合わせることによって，構造物全体の剛性マトリックスを作成する。



図 4.1　要素番号と節点番号

有限要素法では，まず，図のように，節点と要素にそれぞれ番号を付ける。この例題の場合，6 節点，5 要素となる。いま，この要素を骨組要素とすると，各節点に(u_i, v_i, v_i')の 3 つの自由度がある。したがって，この場合の全自由度は,

$$
（節点総数）\times（1 節点の自由度）= 6\times 3 = 18 \tag{4.1}
$$

となる。したがって，構造物全体の剛性マトリックスは図 4.2 ような形になる。図の上側と右側の数字は，1 桁目が節点番号，次が自由度番号を表している。まず，要素①を考えると，要素①の節点番号は 1 と 4 であるから，要素①の剛性マトリックスは，全体剛性マトリックス内の記号 1 の部分に入る。要素②の節点番号は, 2 と 5 であるから, マトリックス内の記号 2 の部分に入る。同様に，要素③，④，⑤は，マトリックス内の記号 3，4，5 の部分に入る。そして，マトリックス内で数字が重なるところは剛性を足し合わせることになる。

これを自動的に配分するためには，まず，全体マトリックスの全成分を 0 にする。次に，各要素の剛性マトリックスの行と列を i 端の節点に関係する部分と j 端の節点に関係する部分に分ける。

$$
\begin{bmatrix} \left[k^{ii}\right] & \left[k^{ij}\right] \\ \left[k^{ji}\right] & \left[k^{jj}\right] \end{bmatrix} \tag{4.2}
$$

ここに,

$$
\left[k_{ii}\right]=\begin{bmatrix} k_{11} & k_{12} & k_{13} \\ k_{21} & k_{22} & k_{23} \\ k_{31} & k_{32} & k_{33} \end{bmatrix},\ \left[k_{ij}\right]=\begin{bmatrix} k_{14} & k_{15} & k_{16} \\ k_{24} & k_{25} & k_{26} \\ k_{34} & k_{35} & k_{36} \end{bmatrix}
$$
$$
\left[k_{ji}\right]=\begin{bmatrix} k_{41} & k_{42} & k_{43} \\ k_{51} & k_{52} & k_{53} \\ k_{61} & k_{62} & k_{63} \end{bmatrix},\ \left[k_{ij}\right]=\begin{bmatrix} k_{44} & k_{45} & k_{46} \\ k_{54} & k_{55} & k_{56} \\ k_{64} & k_{65} & k_{66} \end{bmatrix} \tag{4.3}
$$

そして，これらの 4 つの部分マトリックスを，要素両端の節点番号にしたがって，全体マトリックスに加えていく。

11	12	13	21	22	23	31	32	33	41	42	43	51	52	53	61	62	63	
1	1	1							1	1	1							11
1	1	1							1	1	1							12
1	1	1							1	1	1							13
			2	2	2							2	2	2				21
			2	2	2							2	2	2				22
			2	2	2							2	2	2				23
						3	3	3							3	3	3	31
						3	3	3							3	3	3	32
						3	3	3							3	3	3	33
1	1	1							14	14	14	4	4	4				41
1	1	1							14	14	14	4	4	4				42
1	1	1							14	14	14	4	4	4				43
			2	2	2				4	4	4	245	245	245	5	5	5	51
			2	2	2				4	4	4	245	245	245	5	5	5	52
			2	2	2				4	4	4	245	245	245	5	5	5	53
						3	3	3				5	5	5	35	35	35	61
						3	3	3				5	5	5	35	35	35	62
						3	3	3				5	5	5	35	35	35	63

図 4.2　全体剛性マトリックス

また，全体座標系で定義された要素の剛性マトリックスは，各要素の材料定数（ヤング係数ポアソン比），断面定数（断面積，断面２次モーメント），方向余弦等を与えることによって作成できる。一般に，材料の種別数，断面の種別数は，要素の数に比較して少ないので，通常これらは，要素に付属した情報（材料・断面の種別番号）として与えておく。なお，ここでは簡単のため材料と断面の種別番号は別個にしないで一つの番号で与えるものとする。また, 要素の方向余弦は, 一般に要素両端節点の座標値から計算される。したがって，入力データとしては，各節点の座標値を与えるものとする。

以上の過程を Fortran 言語を用いて記述すると，以下のようになる。

```
      Parameter(mnod=1000,mnel=1000,mnnd=2,mndg=3,mnmt=100)
      Dimension SKG(mnod*mndg,mnod*mndg),SKL(mnnd*mndg,mnnd*mndg)
     1         ,x(mnod),y(mnod)
     1         ,Eyg(mnmt),Are(mnmt),SIz(mnmt)
     1         ,indv(mnel,mnnd),mte(mnel)
c
```

```
      nod = 6                 !節点数
      nel = 5                 !要素数
      nnd = 2                 !一要素の節点数
      ndg = 3                 !一節点の自由度数
      ndgt = ndg*nod          !全自由度数
c
c 要素両端の節点番号
c
      indv(1,1) = 1           !要素１のｉ端の節点番号
      indv(1,2) = 4           !要素１のｊ端の節点番号
      indv(2,1) = 2           !要素２のｉ端の節点番号
      indv(2,2) = 5           !要素２のｊ端の節点番号
      indv(3,1) = 3           !要素３のｉ端の節点番号
      indv(3,2) = 6           !要素３のｊ端の節点番号
      indv(4,1) = 4           !要素４のｉ端の節点番号
      indv(4,2) = 5           !要素４のｊ端の節点番号
      indv(5,1) = 5           !要素５のｉ端の節点番号
      indv(5,2) = 6           !要素５のｊ端の節点番号
c
c 材料・断面種別総数と各要素の材料・断面種別番号の入力
      nmt = 2                 !材料・断面種別数
      mte(1) = 1              !①要素の材料・断面種別番号
      mte(2) = 1              !②要素の材料・断面種別番号
      mte(3) = 1              !③要素の材料・断面種別番号
      mte(4) = 2              !④要素の材料・断面種別番号
      mte(5) = 2              !⑤要素の材料・断面種別番号
c
c 材料定数，断面定数の入力
      Eyg(1) = 2100.d0        !種別番号１のヤング係数
      Are(1) = 100.d0         !種別番号１の断面積
      SIz(1) = 1.d4           !種別番号１の断面２次モーメント
      Eyg(2) = 2100.d0        !種別番号２のヤング係数
      Are(2) = 200.d0         !種別番号２の断面積
      SIz(2) = 2.d4           !種別番号２の断面２次モーメント
c
c 節点座標の入力
      x(1) = 0.d0             !節点１のｘ軸座標
      y(1) = 0.d0             !節点１のｙ軸座標
      x(2) = 90.d0            !節点２のｘ軸座標
      y(2) = 0.d0             !節点２のｙ軸座標
      x(3) = 200.d0           !節点３のｘ軸座標
      y(3) = 0.d0             !節点３のｙ軸座標
      x(4) = 0.d0             !節点４のｘ軸座標
      y(4) = 100.d0           !節点４のｙ軸座標
      x(5) = 90.d0            !節点５のｘ軸座標
      y(5) = 100.d0           !節点５のｙ軸座標
      x(6) = 200.d0           !節点６のｘ軸座標
      y(6) = 100.d0           !節点６のｙ軸座標
c
c 全体剛性マトリックスのゼロクリアー
c
      do 10 i = 1,ndgt
      do 10 j = 1,ndgt
   10 SKG(i,j) = 0.d0
c
c 要素剛性マトリックスの重ね合わせ
c
      do 100 n = 1,nel
      call Elmat(x,y,Eyg,Are,SIz,indv,mte,SKL,n,mnel,mnnd,mndg)
      do 110 ip = 1,nnd
      do 110 i  = 1,ndg
      iL = ndg*( ip-1 ) + i
      iG = ndg*( indv(n,ip)-1 ) + i
      do 120 jp = 1,nnd
      do 120 j  = 1,ndg
      jL = ndg*( jp-1 ) + j
      jG = ndg*( indv(n,jp)-1 ) + j
      SKG(iG,jG) = SKG(iG,jG) + SKL(iL,jL)
```

```
  120 continue
  110 continue
  100 continue
      stop
      end
c
c 要素剛性マトリックスの作成
c
      subroutine Elmat(x,y,Eyg,Are,SIz,indv,mte,SKL,n,mnel,mnnd,mndg)
      dimension SKL(mnnd*mndg,1),x(1),y(1)
     1         ,Eyg(1),Are(1),SIz(1),indv(mnel,1),mte(1)
      m   = mte(n)                        !材料・断面種別番号
      E   = Eyg(m)                        !種別番号 m のヤング係数
      EA  = E*Are(m)                      !種別番号 m のヤング係数×断面積
      EI  = E*SIz(m)                      !種別番号 m のヤング係数×断面２次モーメント
      Ni  = indv(n,1)                     !要素 n の i 端の節点番号
      Nj  = indv(n,2)                     !要素 n の j 端の節点番号
      Xij = x(Nj)-x(Ni)                   ! j 端の x 方向座標 - i 端の x 方向座標
      Yij = y(Nj)-y(Ni)                   ! j 端の y 方向座標 - i 端の y 方向座標
      EL  = sqrt( Xij**2+Yij**2 )         !要素 n の要素長さ
      cs  =  Xij/EL                       !要素 n の cosα
      sn  =  Yij/EL                       !要素 n の sinα
      cs2 = cs**2                         !要素 n の cos²α
      sn2 = sn**2                         !要素 n の sin²α
c
c 要素剛性マトリックスの作成
      SKL(1,1) = EA/EL*cs2 + 12.d0*EI/EL**3*sn2
      SKL(2,1) = (EA/EL-12.d0*EI/EL**3)*cs*sn
      SKL(2,2) = EA/EL*sn2 + 12.d0*EI/EL**3*cs2
      SKL(3,1) =-6.d0*EI/EL**2*sn
      SKL(3,2) = 6.d0*EI/EL**2*cs
      SKL(3,3) = 4.d0*EI/EL
      SKL(4,1) = -SKL(1,1)
      SKL(4,2) = -SKL(2,1)
      SKL(4,3) = -SKL(3,1)
      SKL(4,4) =  SKL(1,1)
      SKL(5,1) = -SKL(2,1)
      SKL(5,2) = -SKL(2,2)
      SKL(5,3) = -SKL(3,2)
      SKL(5,4) =  SKL(2,1)
      SKL(5,5) =  SKL(2,2)
      SKL(6,1) =  SKL(3,1)
      SKL(6,2) =  SKL(3,2)
      SKL(6,3) =  SKL(3,3)/2.d0
      SKL(6,4) = -SKL(3,1)
      SKL(6,5) = -SKL(3,2)
      SKL(6,6) =  SKL(3,3)
c
c 対称成分の代入
      Do 10 i = 1,6
      Do 10 j = i+1,6
   10 SKL(i,j) = SKL(j,i)
c
      return
      end
```